Facultad de Economía, UPC

Taller de modelos macroeconómicos en Matlab

Clase 3

Mg. Carlos Rojas Quiroz

www.carlos-rojas-quiroz.weebly.com

pcefcroj@upc.edu.pe

10 de agosto de 2018

- Construir una curva OA que tenga pendiente positiva para que los choques de demanda (principalmente monetarios) tengan efectos reales.
- Esquema de fijación de precios (competencia monopolística). Fricciones nominales en el mercado de bienes.

Evidencia frecuencia de cambios de precios

		frecuencia (%)		duración[a] (meses)	
Estudio	País	promedio	mediana	promedio	mediana
Auchermanne y Dhyne (2004)	Bélgica	16.9	13.3	5.4	7.0
Nunes (2006)	Brasil	40.3		1.9	
Dhyne *et al* (2006)	Europa	15.1		6.1	
Baudry *et al* (2004)	Francia	18.9	14.9	4.8	6.2
Baharad y Eden (2004)	Israel	24.0	21.0	3.6	4.2
Bils y Klenow (2004)	EE.UU.	26.1	20.9	3.3	4.3
Klenow y Kryvstov (2004)	EE.UU.	29.3		2.9	
Nakamura y Steinsson (2006)	EE.UU.	21.1	8.7	4.2	11.0
Gagnon (2005)	México	30.4 - 36.6		2.2 - 2.8	
Medina et al (2007)	Chile	46.1	33.3	1.6	2.5

Índice

Se hacen importantes inferencias sobre política monetaria utilizando un modelo pequeño (3 ecuaciones):

$$x_t = \mathbb{E}_t(x_{t+1}) - \Psi(i_t - \mathbb{E}_t\pi_{t+1}) + g_t \tag{1}$$

$$\pi_t = \beta\mathbb{E}_t(\pi_{t+1}) + \lambda x_t + u_t \tag{2}$$

$$i_t = \gamma_\pi\mathbb{E}_t(\pi_{t+1}) + \frac{1}{\Psi}g_t \tag{3}$$

El modelo se completa con los procesos AR(1) para los choques de demanda y de oferta:

$$g_t = \mu g_{t-1} + \hat{g}_t \tag{4}$$

$$u_t = \rho u_{t-1} + \hat{u}_t \tag{5}$$

Donde $\hat{g}_t \sim N(0, \sigma_g^2)$ y $\hat{u}_t \sim N(0, \sigma_u^2)$.

- La ecuación 1 es la demanda agregada, donde x_t es la brecha de producto, i_t la tasa de interés nominal y π_t la inflación. Además g_t es un choque de demanda con un proceso AR(1).
- La ecuación 2 es la ecuación de Oferta Agregada o Curva de Phillips. Note que ahora hay una relación positiva entre precios y nivel de actividad (OA con pendiente positiva).
- La ecuación 3 es la regla de Taylor obtenida mediante discreción. Donde $\gamma_\pi = 1 + \frac{(1-\rho)\lambda}{\rho\Psi\alpha}$ (se cumple el principio de Taylor).

La restricción presupuestaria es:

$$W_t + A_t(1 + i_{t-1}) = P_t C_t + A_{t+1} \tag{6}$$

El individuo recibe un salario nominal W_t cada período que lo usa para consumir o acumular activos A_t. A inicios del período t, el individuo posee A_t en activos nominales que pagan una tasa de interés i_{t-1} (que fue pactado a finales del período $t-1$).

Lagrangiano en valor presente:

$$\ell = \sum_{s=t}^{\infty} \beta^{s-t}[U(C_s) + \lambda_s(W_s + A_s(1 + i_{s-1}) - P_s C_s - A_{s+1})] \qquad (7)$$

CPO's:

$$[C_t] : U'(C_t) - \lambda_t P_t = 0 \qquad (8)$$

$$[A_{t+1}] : -\lambda_t + \beta\lambda_{t+1}(1 + i_t) = 0 \qquad (9)$$

Reordenando las CPO's, llegamos a la ecuación de Euler:

$$\frac{U'(C_t)}{P_t} = \frac{1}{1+\rho} \frac{U'(C_{t+1})}{P_{t+1}}(1 + i_t) \qquad (10)$$

Suponiendo una forma funcional específica para la utilidad: $U(C_s) = \frac{C_s^{1-\sigma}}{1-\sigma}$, entonces la ecuación 10 se convierte en:

$$C_t^{-\sigma} = \frac{(1 + i_t)}{1+\rho} \mathbb{E}_t \left\{ C_{t+1}^{-\sigma} \frac{P_t}{P_{t+1}} \right\} \qquad (11)$$

Aplicando logaritmos a la ecuación 11:

$$c_t = \mathbb{E}_t c_{t+1} - \frac{1}{\sigma}(i_t - \mathbb{E}_t \pi_{t+1} - \rho) \tag{12}$$

SI $Y_t = C_t + R_t$, donde R_t es el resto del gasto agregado de la economía, suponemos que $R_t = \chi_t Y_t$. Por tanto:

$$c_t = \log(1 - \chi_t) + y_t \tag{13}$$

Definimos $z_t = -log(1 - \chi_t)$, entonces $y_t = c_t + z_t$, siendo z_t es un choque de demanda AR(1):

$$z_t = \psi z_{t-1} + \varpi_t \tag{14}$$

Donde $\varpi_t \sim N(0, \sigma_z^2)$. Por tanto, $\mathbb{E}_t z_{t+1} = \psi z_t$, entonces:

$$\boxed{y_t = \mathbb{E}_t y_{t+1} + (1 - \psi) z_t - \frac{1}{\sigma}(i_t - \mathbb{E}_t \pi_{t+1} - \rho)} \tag{15}$$

Para pasar de la ecuación 15 a una ecuación de DA en términos de la brecha de producto, podemos incluir el PBI tendencial, $\bar{y}_t$ en ambos miembros:

$$y_t - \bar{y}_t = \mathbb{E}_t y_{t+1} - \mathbb{E}_t \bar{y}_{t+1} + (1-\psi) z_t - \frac{1}{\sigma}(i_t - \mathbb{E}_t \pi_{t+1} - \rho) + \mathbb{E}_t \bar{y}_{t+1} - \bar{y}_t \tag{16}$$

$$x_t = \mathbb{E}_t x_{t+1} - \underbrace{\frac{1}{\sigma}}_{\Psi} (i_t - \mathbb{E}_t \pi_{t+1}) + \underbrace{(1-\psi) z_t + \frac{\rho}{\sigma} + \mathbb{E}_t \Delta \bar{y}_{t+1}}_{g_t} \tag{17}$$

$$\boxed{x_t = \mathbb{E}_t x_{t+1} - \Psi(i_t - \mathbb{E}_t \pi_{t+1}) + g_t} \tag{18}$$

- Estándar en modelos teóricos con rigideces de precios, pues resuelve problemas de agregación y permite ser incorporado en modelos de equilibrio general.
- Las empresas fijan sus precios y ellos permanecen fijos hasta que reciben una señal para cambiarlos.
- El proceso de llegada de esta señal es Poisson, con una probabilidad ψ.
- En cada período *t* habrá algunas firmas cambiando sus precios, ψ, y otra fracción que sigue con ellos fijos, $1 - \psi$.
- Los precios, por lo tanto, serán “traslapados”, es decir, las empresas cambian sus precios en períodos distintos.

- El problema de la firma i-ésima a la que le corresponde cambiar de precio en t es escoger el precio p_{it}. Este precio puede cambiar el siguiente período con probabilidad ψ.
- En el modelo de Calvo, esta probabilidad es exógena.
- En el período t el precio óptimo para la firma es p_t^* , igual que para todas las firmas.
- Asumiendo una función de pérdida cuadrática, el problema de la firma es:

$$\min_{p_{i,t}} C_t = \mathbb{E}_t \left(\sum_{\tau=t}^{\infty} [(1-\psi)\beta]^{\tau-t} (p_{i,t} - p_\tau^*)^2 \right)$$

- $p_{i,t}$ es el precio que fija la firma i en el período t, p_τ^* es el nivel de precios óptimo agregado en el período τ.

$$\frac{\partial C_t}{p_{i,t}} = 2(p_{i,t} - p_t^*) + 2(1-\psi)\beta\mathbb{E}_t(p_{i,t} - p_{t+1}^*) + 2(1-\psi)^2\beta^2\mathbb{E}_t(p_{i,t} - p_{t+2}^*) + \ldots = 0$$

$$p_{i,t}\sum_{j=0}^{\infty}[\beta(1-\psi)]^j - \sum_{j=0}^{\infty}[\beta(1-\psi)]^j\mathbb{E}_t(p_{t+j}^*) = 0$$

FIJACIÓN DE PRECIOS

$$p_{i,t} = (1-\beta(1-\psi))\sum_{j=0}^{\infty}[\beta(1-\psi)]^j\mathbb{E}_t(p_{t+j}^*)$$

Si $\psi = 1$ hay plena flexibilidad de precios. Si $\psi = 0$ hay precios completamente rígidos. A mayor probabilidad, menor ponderación de los períodos futuros.

Descomponiendo el lado derecho de la ecuación:

$$p_{i,t} = (1 - \beta(1-\psi))(p_t^* + \sum_{j=1}^{\infty}[\beta(1-\psi)]^j \mathbb{E}_t(p_{t+j}^*))$$

Tomando el segundo componente del lado derecho:

$$= (1 - \beta(1-\psi))(\sum_{j=1}^{\infty}[\beta(1-\psi)]^j \mathbb{E}_t(p_{t+j}^*))\frac{\beta(1-\psi)}{\beta(1-\psi)}$$

$$= (1 - \beta(1-\psi))(\sum_{j=1}^{\infty}[\beta(1-\psi)]^{j-1} \mathbb{E}_t(p_{t+j}^*))\beta(1-\psi)$$

$$= (1 - \beta(1-\psi))(\sum_{j=0}^{\infty}[\beta(1-\psi)]^{j} \mathbb{E}_t(p_{t+j+1}^*))\beta(1-\psi)$$

Para terminar de resolver el segundo componente del lado derecho, llevamos un período adelante la definición de $p_{i,t}$:

$$\mathbb{E}_t(p_{i,t+1}) = (1-\beta(1-\psi))\sum_{j=0}^{\infty}[\beta(1-\psi)]^j\mathbb{E}_t(p^*_{t+j+1})$$

Reemplazando en lo obtenido hasta el momento:

$$= \underbrace{(1-\beta(1-\psi))(\sum_{j=0}^{\infty}[\beta(1-\psi)]^j\mathbb{E}_t(p^*_{t+j+1}))}_{\mathbb{E}_t(p_{i,t+1})}(\beta(1-\psi)$$

Por lo que llegamos a la expresión:

$$(1-\beta(1-\psi))\sum_{j=1}^{\infty}[\beta(1-\psi)]^j\mathbb{E}_t(p^*_{t+j}) = \beta(1-\psi)\mathbb{E}_t(p_{i,t+1})$$

- Considerando toda la ecuación del precio individual:

$$p_{i,t} = (1-\beta(1-\psi))p_t^* + \beta(1-\psi)\mathbb{E}_t(p_{i,t+1})$$

- Siendo $\mathbb{E}_t(p_{i,t+1})$ el valor esperado de precios futuros corregidos por probabilidad de cambio.
- La ley de movimiento del nivel de precios agregado p_t corresponde a un promedio ponderado entre los precios que fijan las empresas que pudieron cambiar sus precios en t y los precios que traen del período anterior las empresas que no pudieron cambiarlos:

$$p_t = \psi p_{i,t} + (1-\psi)p_{t-1}$$

$$p_t = \psi((1-\beta(1-\psi))p_t^* + \beta(1-\psi)\mathbb{E}_t(p_{i,t+1})) + (1-\psi)p_{t-1}$$

- Falta determinar el precio óptimo p_t^* y la expectativa de precios futuros $\mathbb{E}_t(p_{i,t+1})$. Para este ejemplo se asume:

$$p_t^* = p_t + \phi(y_t - \bar{y}_t) + \vartheta_t$$

- $\vartheta_t \sim N(0, \sigma_\vartheta^2)$. Luego, para $\mathbb{E}_t(p_{i,t+1})$:

$$p_t = \psi p_{i,t} + (1-\psi)p_{t-1}$$

$$p_{i,t} = \frac{1}{\psi}p_t - \frac{(1-\psi)}{\psi}p_{t-1}$$

$$\mathbb{E}_t(p_{i,t+1}) = \frac{1}{\psi}\mathbb{E}_t(p_{t+1}) - \frac{(1-\psi)}{\psi}p_t$$

$$p_t = \psi((1-\beta(1-\psi))(p_t+\phi(y_t-\bar{y}_t)+\vartheta_t)+\beta(1-\psi)(\mathbb{E}_t(\pi_{t+1})+\psi p_t)+(1-\psi)p_{t-1}$$

De aquí se obtiene la curva de Phillips neokeynesiana:

$$p_t = p_{t-1} + \lambda(y_t - \bar{y}_t) + \beta\mathbb{E}_t(p_{t+1} - p_t) + u_t$$

$$\boxed{\pi_t = \lambda x_t + \beta\mathbb{E}_t(\pi_{t+1}) + u_t}$$

Donde:

$$\lambda = \frac{\phi\psi(1-(1-\psi)\beta)}{(1-\psi)}$$

$$u_t = \frac{\psi(1-(1-\psi)\beta)\vartheta_t}{(1-\psi)}$$

Mientras $\psi \to 1$, más vertical es la Curva de Phillips.

Debilidades:

- El modelo no presenta inercia inflacionaria. Se puede introducir componente inercial pero, obviamente, es más compleja la solución.
- La optimización no está en la fuente de la rigidez.

Proviene de un proceso de optimización de la función de pérdida del banco central:

$$\max -\frac{1}{2}[x_t^2 + \pi_t^2] - \underbrace{\frac{1}{2}\mathbb{E}_t \left\{ \sum_{i=1}^{\infty} \beta^i [\alpha x_{t+i}^2 + \pi_{t+i}^2] \right\}}_{F_t} \qquad (19)$$

Sujeto a:

$$\pi_t = \lambda x_t + \underbrace{\beta \mathbb{E}_t \pi_{t+1} + u_t}_{f_t} \qquad (20)$$

Donde α "mide" las preferencias del banco central por la estabilización de la brecha de producto respecto a la estabilización de precios.

La CPO obtenida es:

$$x_t = -\frac{\lambda}{\alpha}\pi_t \tag{21}$$

"*the central bank pursue a "lean against the wind" policy*"

Mientras que la "solución" de la brecha de producto y de la inflación es:

$$x_t = -\lambda q u_t \tag{22}$$

$$\pi_t = \alpha q u_t \tag{23}$$

Donde $q = \frac{1}{\lambda^2 + \alpha(1-\beta\rho)}$. La ecuación 21 se introduce en la DA y se despeja para la tasa de interés nominal con el fin de obtener la Regla de Taylor óptima.

$$\boxed{i_t = \gamma_\pi \mathbb{E}_t \pi_{t+1} + \frac{1}{\Psi} g_t} \tag{24}$$



ygap
x 10^-3
inom
x 10^-3
pic
x 10^-3
u


inom
0.012
0.01
0.008
0.006
0.004
0.002
0
2
4
6
8
10
12
14
16
18
20
g
0.012
0.01
0.008
0.006
0.004
0.002
0
2
4
6
8
10
12
14
16
18
20

Regla de Taylor óptima

La ecuación 24 es obtenida mediante discreción. En el paper de referencia también podemos llegar a obtener resultados mediante compromiso.

- Compromiso con regla simple (restringida).
- Compromiso sin restricción.

El Dynare puede calcular reglas óptimas bajo discreción y bajo compromiso (sin restricción) e inclusive añadir reglas simples óptimas.

Índice

```
var ygap inom pic g u;
varexo ghat uhat;

parameters phi sigma psi beta lambda gamma_pic Psi rho mu;
phi          =0.60;
alpha        =0.50;
sigma        =1.00;
psi          =0.75;
beta         =0.99;
rho          =0.75;
mu           =0.75;
lambda       =phi*psi*(1-(1-psi)*beta)/(1-psi);
Psi          =1/sigma;
gamma_pic    =1+(1-rho)*lambda/(rho*Psi*alpha);
q            =1/(lambda^2+alpha*(1-beta*rho));

model(linear);
ygap     =ygap(+1)-Psi*(inom-pic(+1))+g;
pic      =beta*pic(+1)+lambda*ygap+u;
inom     =gamma_pic*pic(+1)+1/Psi*g;
g        =mu*g(-1)+ghat;
u        =rho*u(-1)+uhat;
end;

shocks;
var ghat; stderr 0.01;
var uhat; stderr 0.01;
end;

stoch_simul(order = 1, nograph, irf=40);
```

OJO: note que en el bloque del modelo no incluimos a nuestro instrumento.

```
model(linear);
ygap      =ygap(+1)-Psi*(inom-pic(+1))+g;
pic       =beta*pic(+1)+lambda*ygap+u;
g         =mu*g(-1)+ghat;
u         =rho*u(-1)+uhat;
end;

shocks;
var ghat; stderr 0.01;
var uhat; stderr 0.01;
end;

planner_objective pic^2 + alpha*ygap^2;

discretionary_policy(planner_discount=0.99,instruments=(
    inom), nograph, irf=40);
```

OJO: similar que con *discretionary_policy*.

```
model(linear);
ygap     =ygap(+1)-Psi*(inom-pic(+1))+g;
pic      =beta*pic(+1)+lambda*ygap+u;
g        =mu*g(-1)+ghat;
u        =rho*u(-1)+uhat;
end;

shocks;
var ghat; stderr 0.01;
var uhat; stderr 0.01;
end;

planner_objective pic^2 + alpha*ygap^2;

ramsey_policy(planner_discount=0.99,instruments=(inom),
    nograph, irf=40);
```



PBI
Inflación
Tasa de interés
Choque
Desv. % EE
Discreción
Compromiso

OJO: en el bloque del modelo se incluye la tasa de interés.

```
model(linear);
ygap     =ygap(+1)-Psi*(inom-pic(+1))+g;
pic      =beta*pic(+1)+lambda*ygap+u;
inom     =gamma_osr*pic(+1)+1/Psi*g;
g        =mu*g(-1)+ghat;
u        =rho*u(-1)+uhat;
end;

shocks;
var ghat; stderr 0.01;
var uhat; stderr 0.01;
end;

optim_weights;
pic  1;
ygap alpha;
end;

osr_params gamma_osr;
gamma_osr=1.5;
osr(nograph, irf=40);
```

Mide el grado de disyuntiva que enfrenta la autoridad monetaria: $\Uparrow \sigma_\pi^2 \rightarrow \Downarrow \sigma_x^2$. Se puede implementar en el Dynare con un *loop* simple en el mismo archivo:

```
nn=[0.0:0.01:1.0];
for j=1:length(nn),
alpha=nn(1,j);

planner_objective pic^2 + alpha*ygap^2;

discretionary_policy(planner_discount=0.99,instruments=(
    inom), nograph, irf=40);
var_ygap(j,1)      =oo_.var(1,1);
var_pic(j,1)       =oo_.var(3,3);
end;
varianzas =[var_pic*10000 var_ygap*10000];
scatter(varianzas(:,1), varianzas(:,2))
xlabel('Varianza Inflacion','Fontsize',12)
ylabel('Varianza Brecha PBI','Fontsize',12)
```

Frontera de Política



Índice

No es necesario incluir microfundamentos, sino formular ecuaciones que sean empíricamente significativas, aunque con lógica económica detrás. Podemos reformular nuestro sistema de tres ecuaciones:

$$x_t = \gamma_x x_{t-1} + (1-\gamma_x)\mathbb{E}_t x_{t+1} - \Psi(i_t - \mathbb{E}_t \pi_{t+1}) + g_t \tag{25}$$

$$\pi_t = \alpha_\pi \pi_{t-1} + (1-\alpha_\pi)\mathbb{E}_t \pi_{t+1} + \lambda x_t + u_t \tag{26}$$

$$i_t = \rho_i i_{t-1} + (1-\rho_i)(\phi_x x_t + \phi_\pi \pi_t) + z_t \tag{27}$$

Observe que incluímos componentes *forward* y *backward looking* en el modelo. Además, note que **la regla de Taylor contiene un componente inercial** y que depende del valor actual de la brecha de producto y de la inflación. Además, z_t es una "sorpresa monetaria".

Choque de demanda




PBI
Desv. % EE
Inflación
Desv. % EE
Tasa de interés
Desv. % EE
Choque
Desv. % EE
γ_x=0, α_π=0
γ_x=0.45, α_π=0.45


PBI
Desv. % EE
0.5
0
-0.5
-1
-1.5
-2
-2.5
2
4
6
8
10
12
14
16
18
20
Inflación
0
-1
-2
-3
-4
-5
Tasa de interés
-0.2
-0.4
-0.6
-0.8
-1.2
-1.4
-1.6
-1.8
Choque
1.4
1.2
1
0.8
0.6
0.4
0.2
γ_x=0, α_π=0
γ_x=0.45, α_π=0.45